
Epsilon macchina:

real function eps()

c

c  SCOPO: calcolo dell'epsilon macchina di un sistema floating-point.

c        

c  PARAMETRI DI INPUT:  

c 

c  - nessuno

c

c  PARAMETRI DI OUTPUT:

c

c  - eps,  reale,  epsilon macchina

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna 

c

       real e,e1,e2        

       e1=1

       e2=e1+1

c si ripete il ciclo finchè la variabile e2 

c non da contributo alla somma.      

   10  continue  

       if (e2 .eq. 1) goto 20

          e=e1

          e1=e1/2

          e2=e1+1

       goto 10

   20  continue

       eps=e

      End     

Minimo numero rappresentabile:

real function minr()

c

c  SCOPO: calcolo del minimo numero positivo di un sistema 

c         floating-point.

c        

c  PARAMETRI DI INPUT:  

c 

c  - nessuno

c

c  PARAMETRI DI OUTPUT:

c

c  - minr,  reale,  minimo numero positivo rappresentabile

c   

c  ROUTINES RICHIAMATE:

c

c  - eps,  funzione reale,  epsilon macchina 

c            

       real eps

       external eps

       real xold,xnew

       xnew=1.0

   10  if (xnew.eq.0) goto 20

          xold=xnew

          xnew=xold/2

       goto 10

   20  continue

c     Operazione di shift left della mantissa al fine di

c     ottenere il minimo rappresentabile del sistema

c     floating-point normalizzato. Non normalizzato il

c     minimo numero rappresentabile è circa 10^-45

       minr=xold/(2*eps())

      End

Tolleranza minima:

real function tolm(x)

c

c  SCOPO: calcolo della tolleranza minima, relativa ad un numero x

c         di un sistema floating-point.

c        

c  PARAMETRI DI INPUT:  

c 

c  - x,  reale,  numero di cui calcolare la tolleranza relativa

c

c  PARAMETRI DI OUTPUT:

c

c  - tolm,  reale,  tolleranza minima reltiva ad x

c   

c  ROUTINES RICHIAMATE:

c

c  -  eps,  funzione reale,  epsilon macchina

c  - minr,  funzione reale,  minimo rappresentabile

c

       real eps,minr

       external eps,minr

       real e,m,x

       e=eps()

       m=minr()

       if (abs(x) .gt. m/(2*e)) then

          tolm=abs(x)*2*e

       else 

          tolm=m

       endif

      End   

Funzione esponenziale:

real function espon(x, tol, ifail)

c

c  SCOPO: calcolo dell' esponenziale di un numero. 

c        

c  PARAMETRI DI INPUT:  

c 

c  -   x,  reale,  punto in cui calcolare la funzione

c  - tol,  reale,  tolleranza richiesta dall’utente

c 

c  PARAMETRI DI OUTPUT:

c

c  - espon,   reale,  valore della funzione in x

c  - ifail,  intero,  indicatore d'errore

c   

c  ROUTINES RICHIAMATE:

c

c  - tolm,  funzione reale,  tolleranza minima relativa

c

       real tolm

       external tolm

       real x,tol

       integer ifail

       integer n

c     Controllo della tolleranza        

       if (tol .lt. 0) then

          ifail=1

          goto 999

       endif

       vax=abs(x)

       sum=0

       add=1

       n=0              

   10  continue

          sum=sum+add

          n=n+1

          add=add*(vax/n)

       if (add .ge. (tol*sum+tolm(sum))) goto 10

c      x>0 espon = exp(x)

       if(add .ge. 0) then 

          espon=sum

c      x<0 espon=exp(-x)         

       else

          espon=1/sum

       endif   

       ifail=0                                         

  999  continue    

      End     

Fattorizzazione LU con pivoting parziale simulato:

Subroutine lup(a, lda, n, ipiv, ifail)

c

c  SCOPO: fattorizzazione di una matrice quadrata A nella forma

c         PA=LU, dove P è la matrice delle permutazioni derivante

c         dall' applicazione del pivoting, L una matrice triango-

c         lare inferiore a diagonale unitaria e U triangolare su-

c         periore.

c 

c  PARAMETRI DI INPUT:  

c 

c  -   a,  matrice di reali,  matrice da fattorizzare

c  - lda,            intero,  leading dimension di a dichiarata 

c                             nel programma chiamante

c  -   n,            intero,  ordine della matrice

c 

c  PARAMETRI DI OUTPUT:

c

c  -     a,   matrice di reali,  permutate le righe secondo ipiv

c                                restituisce i fattori L ed U 

c  -  ipiv,  vettore di interi,  indici delle righe di a da per-

c                                mutare per ottenere L ed U

c  - ifail,             intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna.

c

c      |a11 a12 a13 .. a1n|     |l11 0   0   .. 0 |   |u11 u12 u13 .. u1n|

c      |a21 a22 a23 .. a2n|     |l21 l22 0   .. ..|   |0   u22 u23 .. u2n|

c  A = |a31 a32 a33 .. a3n|  =  |l31 l32 l33 .. ..| * |0   0   u33 .. u3n|

c      |..  ..  ..  .. .. |     |..  .. ..  .. ..|   |..  ..  .. .. .. | 

c      |an1 an2 an3 .. ann|     |ln1 ln2 ln3 .. lnn|   |0   ..   .. .. unn|

c                                          

c                                   L                    U

c

       integer lda, n, ipiv(n), ifail

       real    a(lda,n)

c     Controllo della dimensione 

       if (n .le. 0 .or. n .gt. lda) then

          ifail=3

          goto 999

       endif

c     Inizializzazione di ipiv

       do 10 i=1,n

          ipiv(i)=i

10  continue

do 60 k=1,n

c     Ricerca del pivot

c     p, il pivot è il massimo elemento della riga k, colonna k

          p=abs(a(ipiv(k),k))                                  

c     m è uguale all'indice di riga del pivot          

          m=k                            

c     cerca nelle righe sottostanti k, un pivot maggiore in modulo

c     e se lo trova, diventa il nuovo pivot

          do 20 i=k+1,n

             if (abs(a(ipiv(i),k)) .gt. p) then

                p=abs(a(ipiv(i),k))

                m=i



endif  

   20     continue

c     Controllo della singolarità (il pivot è zero)

          if (p .eq. 0) then

             ifail=4

             goto 999

          endif 

c     Scambio virtuale                   

c     il pivot non è sulla riga k, ma sulla m, 

c     le due righe vanno scambiate

          if (m .ne. k) then

             temp=ipiv(m)                   

             ipiv(m)=ipiv(k)                     

             ipiv(k)=temp

          endif

c     Calcolo dei moltiplicatori

c
a11x1+a12x2+a13x3+ … 
c
a21x1+a22x2+a23x3+ … 
c
…

c
an1x1+an2x2+an3x3+ … 
c

c     mi1 = ai1/a11       i=2,3,…,n
- divisione per pivot

c     aij = aij - mikakj


- fattorizzazione

c
divido tutti gli elementi al di sotto

c
del pivot per il pivot stesso

          do 30 i=k+1,n                       

             a(ipiv(i),k)=a(ipiv(i),k)/a(ipiv(k),k)

30 continue  

c
fattorizazione

          do 50 j=k+1,n

             do 40 i=k+1,n 

                a(ipiv(i),j)=a(ipiv(i),j)-a(ipiv(i),k)*a(ipiv(k),j)

   40        continue          

   50     continue

   60  continue

       ifail=0

  999  continue

      End

Soluzione di Ax=b con A=LU (applicando forward e backward substitution):

       Subroutine slup(lu, p, b, ldlu, n, ifail)

c

c  SCOPO: risoluzione di un sistema lineare A*x=b mediante la ri-

c         soluzione del sistema equivalente (L*U)*x=P*b, con P ma-

c         trice delle permutazioni ed L*U la fattorizzazione di 

c         P*A. La backward e forward substitution sono implementate

c         per colonna 

c

c  PARAMETRI DI INPUT:  

c 

c  -   lu,   matrice di reali,  permutate le righe secondo p

c                               restituisce i fattori L ed U 

c  -    p,  vettore di interi,  indici delle righe di a da 

c                               permutare per ottenere L ed U   

c  -    b,   vettore di reali,  termini noti

c  - ldlu,             intero,  leading dimension di lu dichia-

c                               rata nel programma chiamante

c  -    n,             intero,  ordine del sistema 

c

c  PARAMETRI DI OUTPUT:

c

c  -     b,  vettore di reali,  soluzione del sistema

c  - ifail,            intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna.

c

c
Ax=b; A=LU;  (LU)x=b; L(Ux)=b; Ux=y; Ly=b

c

c    |a11  a12  a13  ..  a1n|   |b1|    |b1|

c    |a21  a22  a23  ..  a2n|   |b2|    |b2|

c    |a31  a32  a33  ..  a3n| * |b3|  = |b3|

c    |..   ..   ..   ..  .. |   |..|    |..|

c    |an1  an2  ..   ..  ann|   |b4|    |b4|

c                                          

c    forward substitution                      

c    |  1    0   0  ..   0|   |b1|   |b1| |

c    |l21    1   0  ..  ..|   |b2|   |b2| |

c    |l31  l32   1  ..  ..| * |b3| = |b3| v   Ly = b   

c    |..   ..   ..  ..  ..|   |..|   |..|

c    |ln1  ln2  ..  ..   1|   |bn|   |bn|

c

c    backward substitution 

c    |u11  u12  u13 .. u1n|   |b1|   |b1|

c    |0    u22  u23 .. u2n|   |b2|   |b2|

c    |..   0    u33 .. u3n| * |b3| = |b3| ^   Ux = y

c    |..   ..   ..  ..  ..|   |..|   |..| |

c    |0    ..   ..  .. unn|   |bn|   |bn| |

c                                        

       integer ldlu, n, p(n), ifail

       real    lu(ldlu,n), b(n)

c     Controllo della dimensione

       if (n .le. 0  .or.  n .gt. ldlu) then

         ifail=3

         goto 999

       endif                   

c     Controllo della singolarità

       do i=1,n

          if (lu(p(i),i) .eq. 0) then

             ifail=4

             goto 999

          endif            

       end do

c
forward substitution v. colonna -- L*y = P*b

       do j=1, n-1

          do i=j+1, n

             b(i)=b(i)- lu(p(i),j)*b(j)

          end do

       end do

c
backward substitution v. colonna -- U*x = y

       do j=n,2,-1

          b(j)=b(j)/lu(p(j),j)

          do i=j-1,1,-1

             b(i)=b(i)- lu(p(i),j)*b(j)

          end do

       end do

       b(1)=b(1)/lu(p(1),1)

       ifail=0

  999  continue

      end
Backward substitution formato packed:

Subroutine bsbpk (u, b, n, ifail)

c

c  SCOPO: risoluzione di un sistema lineare U*x=b, con matrice

c         dei coefficienti U di tipo triangolare superiore.

c         La matrice è in formato packed.

c 

c  PARAMETRI DI INPUT:  

c 

c  - u,  vettore di reali,  elementi del triangolo superiore

c  - b,  vettore di reali,  termini noti

c  - n,            intero,  ordine del sistema

c 

c  PARAMETRI DI OUTPUT:

c

c  -     b,  vettore di reali,  soluzione del sistema

c  - ifail,            intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna

c

c       |a11  0    0   …   0|

c       |a21  a22  0   …   0|

c   A = |a31  a32  a33 …   0|    A(i,j) => u(k)  

c       |…    …    …   …   …|    k = (i-1)n – i(i-1)/2 + j

c       |an1  an2  an3 … ann|

c

c   u = |a11 a21 a22 a31 … ann|

c

       integer n, ifail, log

       real    u(n*(n+1)/2), b(n)

c  Controllo della dimensione 

       if (n .le. 0) then 

          ifail=3

          goto 999

       endif

c  Controllo della singolarità 

       id=n*(n+1)/2 + 1                         

       do i=1,n                           

          id=id-i               

          if (u(id) .eq. 0) then

             ifail=4

             goto 999

          endif

       end do

c  
Backward Substitution        

c  
Per accedere all’elemento a(i,j) dovrei calcolare 

c  
k = (i-1)n – i(i-1)/2 + j

c
che richiede 3M+4A a ciclo. La variabile k assumerebbe

c  
i valori da n(n+1)/2+1 a 1, quindi evito di

c  
calcolare 3M+4A a ciclo e simulo il 

c  
comportamento di k con la variabile iu, opportunamente

c  
inizializzata e decrementata ad ogni ciclo.

       iu=n*(n+1)/2 + 1

       do i=n,1,-1

          do j=n,i+1,-1

             iu=iu-1

             b(i)=b(i)-u(iu)*b(j)   

          end do

          iu=iu-1

          b(i)=b(i)/u(iu)

       end do

       ifail=0

  999  continue

       end       

Forward substitution formato packed:

Subroutine fsbpk (l, b, n, ifail)

c

c  SCOPO: risoluzione di un sistema lineare L*x=b, con matrice dei 

c         coefficienti L di tipo triangolare inferiore.

c         La matrice è in formato packed

c 

c  PARAMETRI DI INPUT:  

c 

c  - l,  vettore di reali,  coefficienti non nulli

c  - b,  vettore di reali,  termini noti

c  - n,            intero,  ordine del sistema

c 

c  PARAMETRI DI OUTPUT:

c

c  -     b,  vettore di reali,  soluzione del sistema

c  - ifail,  intero          ,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna

c

c       |a11 a12  a13  …  a1n|

c       | 0  a22  a23  …  a2n|

c   A = | …   0   a33  …  a3n|    A(i,j) => l(k) 

c       | …   …    0   …  …  |    k = j + i(i-1)/2

c       | 0   0    …   …  ann|

c

c   l = |a11 a12 a13 … an-1n ann|

c

       integer n, ifail

       real    l(n*(n+1)/2), b(n)             

c     Controllo della dimensione

       if (n .le. 0) then 

          ifail=3

          goto 999

       endif

c     Controlla della singolarità

       id=0

       do 10 i=1,n                           

          id=id+i

          if (l(id).eq.0) then

             ifail=4

             goto 999

          endif

   10  continue

c     Forward Substitution

c     Per accedere all’elemento a(i,j) dovrei calcolare 

c     k = i(i-1)/2 + j

c  
che richiede 2M+2A a ciclo. La variabile k assumerebbe

c  
i valori da 1 a n(n+1)/2, quindi evito di

c  
calcolare 2M+2A a ciclo e simulo il 

c  
comportamento di k con la variabile il, opportunamente

c  
inizializzata e incrementata ad ogni ciclo.

c
il = j + i(i-1)/2 = k

       il=0

       do 30 i=1,n

          do 20 j=1,i-1     

             il=il+1

             b(i)=b(i)-l(il)*b(j)

   20     continue

          il=il+1

          b(i)=b(i)/l(il)

   30  continue       

       ifail=0  

  999  continue

      End        

Fattorizzazione LU con pivoting scalato:

       Subroutine lups(a, lda, n, ipiv, s, ifail)

c

c  SCOPO: fattorizzazione di una matrice quadrata A nella forma

c         P*A=L*U, dove P è la matrice delle permutazioni derivante

c         dall' applicazione del pivoting scalato, L una matrice

c         triangolare inferiore a diagonale unitaria e U triangolare

c         superiore.

c 

c  PARAMETRI DI INPUT:  

c 

c  -   a,  matrice di reali,  matrice da fattorizzare

c  - lda,            intero,  leading dimension di a dichiarata 

c                             nel programma chiamante

c  -   n,            intero,  ordine della matrice

c 

c  PARAMETRI DI OUTPUT:

c

c  -     a,   matrice di reali,  permutate le righe secondo ipiv

c                                restituisce i fattori L ed U 

c  -  ipiv,  vettore di interi,  indici delle righe di a da per-

c                                mutare per ottenere L ed U

c  -     s,   vettore di reali,  norma infinito delle righe di A

c  - ifail,             intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - eps,  funzione reale,  epsilon macchina

c  

       real eps

       external eps

       integer lda, n, ipiv(n), ifail

       real    a(lda,n), s(n)

c  Controllo della dimensione

       if (n .le. 0 .or. n .gt. lda) then

          ifail=3                        

          goto 999

       endif                             

       e=eps()                                           

c  Inizializzazione di ipiv e ricerca dei massimi di riga 

c  s(i) è il massimo elemento in modulo della riga i

       do 20 i=1,n

          ipiv(i)=i      

          s(i)=a(i,1)

          do 10 j=2,n 

             s(i)=amax1(s(i),abs(a(i,j)))

   10     continue

c     Controllo della singolarità      

          if (s(i) .lt. e) then

             ifail=4                            

             goto 999

          endif

   20  continue                             

       do 70 k=1,n

c     Ricerca e Scalatura del pivot.     

c
si divide il pivot per il massimo elemento

c     in modulo della riga in oggetto 

         ps=abs(a(ipiv(k),k))/s(k)

          m=k

          do 30 i=k+1,n                                       

             as=abs(a(ipiv(i),k))/s(i)   

             if (as .gt. ps) then

                ps=as

                m=i                         

             endif

30 continue

c     Controllo della singolarità

          if (ps .lt. 0) then

             ifail=4                          

             goto 999

          endif                          

c     Scambio virtuale 

          if (m .ne. k) then

             temp=ipiv(m)

             ipiv(m)=ipiv(k)

             ipiv(k)=temp  

          endif

c     Calcolo dei moltiplicatori

          do 40 i=k+1,n

                a(ipiv(i),k)=a(ipiv(i),k)/a(ipiv(k),k)

   40     continue       

c
Fattorizzazione

           do 60 j=k+1,n

             do 50 i=k+1,n  

                a(ipiv(i),j)=a(ipiv(i),j)-a(ipiv(i),k)*a(ipiv(k),j)

   50        continue                                           

   60     continue 

   70  continue

       ifail=0

  999  continue

      End      

Correzione Iterativa:

Subroutine corit(lu, p, b, x, ldlu, n, tol, maxit, r, ifail)

c

c  SCOPO: correzione iterativa della soluzione di un sistema li-

c         neare A*x=b mediante il sistema equivalente (L*U)*x=P*b,

c         dove P è una matrice di permutazione ed L*U la fattoriz-

c         zazione di P*A fornita in uscita dalla subroutine lup.

c

c  PARAMETRI DI INPUT: 

c                                                   

c  -    lu,   matrice di reali,  permutate le righe secondo p

c                                restituisce i fattori L ed U 

c  -     p,  vettore di interi,  indici delle righe di A da 

c                                permutare per ottenere L ed U   

c  -     b,   vettore di reali,  vettore termini noti

c  -     x,   vettore di reali,  soluzione iniziale

c  -   tol,              reale,  tolleranza richiesta dall’utente

c  - maxit,             intero,  massimo numero di iterazioni

c  -  ldlu,             intero,  leading dimension di LU dichiarata

c                                nel programma chiamante

c  -     n,             intero,  ordine del sistema

c

c  PARAMETRI DI OUTPUT:

c

c  -     b,  vettore di reali,  soluzione finale del sistema

c  - ifail,            intero,  indicatore di errore

c

c  ROUTINE RICHIAMATE:

c

c  - nessuna

c

       integer maxit, ldlu, n, p(n), ifail, iter

       real    lu(ldlu,n), b(n), normainf, tol

       double precision x(n), r(n)

c     Controllo della dimensione

      if (n .le. 0 .or. n .gt. ldlu) then

          ifail = 3

          goto 999

      endif

c     controllo della tolleranza

      if (tol .lt. 0) then

          ifail = 1

          goto 999

      endif

c     Controllo sul numero minimo di iterazioni richieste

      if (maxit .lt. 1) then

          ifail = 8

          goto 999

      endif                                    

c     Controllo della singolarità di lu

      do 5 i=1,n

          if (lu(p(i),i) .eq. 0) then 

              ifail = 4

              goto 999

          endif

    5 continue

      iter=0

   10 continue   

      iter=iter+1  

c                     (s)           (s)                   

c     Risoluzione di r     =   b - Ax 

-
s=iter

c 
Il residuo è calcolato in doppia precisione

      do 30 i=1,n

          r(i) = b(i)

          do 20 j = 1,n

              r(i) = r(i) - lu(p(i),j) * x(j)

   20     continue

   30 continue

c     Risoluzione di Adx=r, effettuata applicando la

c     forward e la backward substitution, implementate per colonna; 

c
A*dx=r = L(Udx)=Pr, Udx=y

c      Risoluzione di L*y = P*r 

       do 40 j=1,n

          do 35 i=j+1,n              

             r(i) = r(i) - lu(p(i),j)*r(j)  

   35     continue                

   40  continue

c     Risoluzione di U*x = y

       do 50 j=n,1,-1

          r(j) = r(j) / lu(p(j),j)

          do 45 i=j-1,1,-1             

             r(i) = r(i) - lu(p(i),j)*r(j) 

   45     continue       

   50  continue

c     Calcolo della soluzione approssimante  

      do 60 i=1,n       

          x(i)=x(i)+r(i)  

   60 continue

c     Calcolo della norma infinita di dx(=r)

c 
x = max |xi|
c           1  i n

      normainf=0

      do 70 i=1,n

          if (abs(r(i)) .gt. normainf) then

              normainf = abs(r(i))

          endif

   70 continue

c     controllo sul numero di iterazioni effettuate

      if (iter .eq. maxit) then

          ifail = 9

          goto 999

      endif

c     controllo sul raggiungimento delle condizioni di uscita dal ciclo

      if ((iter .lt. maxit) .or. (normainf .gt. tol)) goto 10

  999 continue   

      end   

Fattorizzazione LU per matrici tridiagonali :

Subroutine lut(d, e, f, n, ifail)

c

c  SCOPO: fattorizzazione di una matrice quadrata tridiagonale A

c         nella forma A=L*U, dove L è una matrice bidiagonale in-

c         feriore a diagonale unitaria e U bidiagonale superiore.

c

c  PARAMETRI DI INPUT:

c                      

c  - d,  vettore di reali,  diagonale principale 

c                           della matrice da fattorizzare

c  - e,  vettore di reali,  diagonale inferiore 

c                           della matrice da fattorizzare

c  - f,  vettore di reali,  diagonale superiore 

c                           della matrice da fattorizzare

c  - n,            intero,  ordine della matrice

c

c  PARAMETRI DI OUTPUT:

c

c  -     d,  vettore di reali,  diagonale principale

c                               della matrice fattorizzata

c  -     e,  vettore di reali,  diagonale inferiore

c                               della matrice da fattorizzata

c  -     f,  vettore di reali,  diagonale superiore

c                               della matrice da fattorizzata

c  - ifail,            intero,  indicatore di errore

c

c  ROUTINE RICHIAMATE:

c

c  - nessuna

c  

c      |a1   c1   0    …   0  |     d = (a1 a2 a3 … an)

c      |b1   a2   c2   …   0  |     e = (b1 b2 … bn-1)

c  A = |0    b2   a3   …   0  |     f = (c1 c2 … cn-1)

c      |…    …    …    …  cn-1|     

c      |0    0    0  bn-1 an  |

c

c      |1   0   .   .   0|       |d1  f1  0   .   .  | 

c      |e1  1   0   .   .|       |0   d2  f2  .   .  | 

c  L = |0   e2  1   .   .|   U = |.   0   d3  .   .  |   

c      |.   .   .   .   .|       |.   .   0   .  fn-1|

c      |0   .   .  en-1 1|       |0   .   .   .   dn |

c

c          |d1    f1       0        .       .        |

c          |e1d1  e1f1+d2  f2       .       .        |

c  A = LU =|0     e2d2     e2f2+f3  f3       .        |

c          |.     0        .      .       fn-1      |

c          |.     .        .      en-1fn-1  en-1fn-1+dn| 

c

c  f(i) = c(i)                                    diagonale superiore

c  d(1) = a(1) 

c  e(i) = b(i)/d(i) => e(i) = e(i)/d(i)           diagonale inferiore

c  d(i+1) = a(i+1)–e(i)f(i) = d(i+1)-e(i)f(i)     diagonale principale

c  i= 1,..,n-1

c

       integer n, ifail

       real    d(n), e(n-1), f(n-1)

c  Controllo della dimensione 

       if (n .le. 0) then

          ifail=3

          goto 999

       endif          

c  Fattorizzazione

       do 10 k=1,n-1

c  Controllo della singolarità di un minore principale

c  non viene controllato se d(n) .eq. 0, ma non occorre

c  in quanto d(n) non è utilizzato per calcolare e(n-1),

c  se anche fosse nullo non si effettua la divisione per zero

          if (d(k) .eq. 0) then

             ifail=6

             goto 999

          endif                                    

c  Calcolo di un elemento della diagonale inferiore          

          e(k)=e(k)/d(k)

c  Calcolo di un elemento della diagonale principale

          d(k+1)=d(k+1)-e(k)*f(k)

   10  continue   

       ifail=0

  999  continue

       End      

Soluzione di un sistema tridiagonale (fattorizzato LU) con back e forward substitution :

Subroutine slut(d, e, f, b, n, ifail)

c

c  SCOPO: risoluzione di un sistema lineare tridiagonale A*x=b me-

c         diante la risoluzione del sistema equivalente (L*U)*x=b,

c         dove L*U è la fattorizzazione di A.

c

c  PARAMETRI DI INPUT:

c

c  - d,  vettore di reali,  diagonale principale di U

c  - e,  vettore di reali,  diagonale inferiore di L

c  - f,  vettore di reali,  diagonale superiore di U 

c  - b,  vettore di reali,  colonna dei termini noti

c  - n,            intero,  ordine del sistema

c

c  PARAMETRI DI OUTPUT:

c

c  -     b,  vettore di reali,  soluzione del sistema

c  - ifail,            intero,  indicatore di errore

c

c  ROUTINE RICHIAMATE:

c  

c  - nessuna.

c

c      |1   0   .   .   0|   |b1|   |b1|

c      |e1  1   0   .   .|   |b2|   |b2|

c  L = |0   e2  1   .   .| * |b2| = |b3|

c      |.   .   .   .   .|   |..|   |..|

c      |0   .   .  en-1 1|   |bn|   |bn|

c

c      |d1  f1  0   .  .  |    |b1|   |b1|

c      |0   d2  f2  .  .  |    |b2|   |b2|

c  U = |.   0   d3  .  .  |  * |b3| = |b3|

c      |.   .   0   . fn-1|    |..|   |..|

c      |0   .   .   .  dn |    |bn|   |bn|

c

       integer n, ifail

       real    d(n), e(2:n), f(n-1), b(n)

c     Controllo della dimensione

       if (n .le. 0) then

          ifail=3

          goto 999

       endif

c     Controllo della singolarità di U

       do 10 i=1,n

          if (d(i) .eq. 0) then

             ifail=4

             goto 999

          endif

   10  continue

c     Risoluzione di L*y=b (con y=U*x)

c     forward substirution v. colonna

       do 20 i=2,n

          b(i)=b(i)-e(i)*b(i-1)

   20  continue

c     Risoluzione di U*x=y 

c     backward substitution v. colonna

       b(n)=b(n)/d(n)

       do 30 i=n-1,1,-1

             b(i)=(b(i)-f(i)*b(i+1))/d(i)

   30  continue

       ifail=0

  999  continue

      End

Fattorizzazione LU per matrici a banda:

Subroutine lub(a, lda, n, p, q, ifail)

c

c  SCOPO: fattorizzazione di una matrice quadrata a banda A nella

c         forma A=L*U, dove L è una matrice triangolare inferiore

c         a diagonale unitaria e U triangolare superiore.

c

c  PARAMETRI DI INPUT:

c

c  -   a,  matrice di reali,  matrice da fattorizzare

c  - lda,            intero,  leading dimension di a dichiarata

c                             nel programma chiamante

c  -   n,            intero,  ordine della matrice

c  -   p,            intero,  ampiezza della banda inferiore

c  -   q,            intero,  ampiezza della banda superiore

c

c  PARAMETRI DI OUTPUT:

c

c  -     a,  intero,  matrice fattorizzata

c  - ifail,  intero,  indicatore di errore

c

c  ROUTINE RICHIAMATE:

c

c  - nessuna

c

c  es.

c                 q

c               ------  

c          |a1  b1  c1   0   0|         | 0   0  c1  c2  c3| \ q

c       p| |d1  a2  b2  c2   0|         | 0  b1  b2  b3  b4| / 

c  A =     | 0  d2  a3  b3  c3| => Ab = |a1  a2  a3  a4  a5| 

c          | 0   0  d3  a4  b4|         |d1  d2  d3  d4   0| - p

c          | 0   0   0  d4  a5|                             

c  

c  A(i,j) => Ab(i-j+q+1,j)

c

       integer n, p, q, ifail, amp

       real    a(lda,n)

       amp=p+q+1

c  Controllo della dimensione

       if (n .le. 0  .or.  n .gt. lda) then

          ifail=3

          goto 999

       endif

c  Controllo delle ampiezze di banda

       if (p .le. 0 .or. q .le. 0 .or. p .ge. n .or. q .ge. n) then

          ifail=6

          goto 999

       endif

c      Fattorizzazione

       do 40 k=1,n-1

          do 10 i=k+1,min(k+p,n)                  

c        Controllo della singolarità di un minore principale             

             if (a(q+1,k) .eq. 0) then

                ifail=7

                goto 999

             endif                 

             ir=i-k+q+1  

c        Calcolo dei moltiplicatori   

             a(ir,k)=a(ir,k)/a(q+1,k)    

   10     continue

          do 30 j=k+1,min(k+q,n)

             do 20 i=k+1,min(k+p,n)

                ir=i-j+q+1      

                a(ir,j)=a(ir,j)-a(i-k+q+1,k)*a(k-j+q+1,j)

   20         continue

   30     continue

   40  continue

       ifail=0

  999  continue

      End

Risoluzione di un sistema lineare con A matrice a banda 

      Subroutine slub(a,b,lda,n,p,q,ifail,log)

c                

c  SCOPO: risoluzione di un sistema lineare Ax=b, con A

c         matrice a banda, di banda inferiore p e banda 

c         superiore q. La matrice A è compattata 

c         secondo la seguente formula:

c         A(i,j) => Al(i-j+q+1,j)

c

c  PARAMETRI DI INPUT:

c

c  -   a,  matrice di reali,  matrice dei coefficienti

c  -   b,  vettore di reali,  termine noto

c  - lda,            intero,  leading dimension di a dichiarata

c                             nel programma chiamante

c  -   n,            intero,  ordine della matrice

c  -   p,            intero,  ampiezza della banda inferiore

c  -   q,            intero,  ampiezza della banda superiore

c

c  PARAMETRI DI OUTPUT:

c

c  -     b, vettore di reali, soluzione del sistema

c  - ifail,           intero, indicatore di errore

c

c  ROUTINE RICHIAMATE:

c

c  - nessuna                 

c

c  es.    

c         |a1  b1  c1   0   0|         | 0   0  c1  c2  c3| \ q

c         |d1  a2  b2  c2   0|         | 0  b1  b2  b3  b4| / 

c     A = | 0  d2  a3  b3  c3| => Ab = |a1  a2  a3  a4  a5|   q+1

c         | 0   0  d3  a4  b4|         |d1  d2  d3  d4   0| - p

c         | 0   0   0  d4  a5|                             

c

c     La matrice A, compattata assume la forma della matrice Al, 

c     la fattorizzazione compiuta su Al, non altera le dimensioni 

c     di banda. La matrice L ottenuta dalla fattorizzazione

c     si troverà nelle ultime p righe di Al, la matrice U, si 

c     troverà nelle prime q righe di Al, la diagonale principale

c     di U si trova alla riga q+1.

c

       integer n, p, q, ifail

       real    a(lda,n), b(n)

c     Controllo della dimensione

       if (n .le. 0  .or.  n .gt. lda) then

          ifail=3

          goto 999

       endif

c     Controllo delle ampiezze di banda

       if (p .le. 0 .or. q .le. 0 .or. p .ge. n .or. q .ge. n) then

          ifail=6

          goto 999

       endif            

c     Controllo sulla singolarità della matrice

      do i=1,n

          if (a(q+1,i) .eq. 0) then

              ifail = 7

              goto 999

          end if

      end do           

c     forward substitution –- L*y = b   

      do j=1,n-1

          do i=j+1, min(j+p,n)

              b(i)=b(i)-a(i-j+q+1,j)

          end do

      end do

c     backward substitution –- U*x = y                                          

      do j=n,2,-1            

          b(j)=b(j)/a(q+1,j)

          do i=j-1, max(1,j-q),-1                        

              b(i)=b(i)-a(i-j+q+1,j)*b(j)

          end do    

      end do

      b(1)=b(1)/a(q+1,1)

       ifail=0

  999  continue

      End   

Fattorizzazione di Cholesky

Subroutine chol(a, l, n, ifail)

c

c  SCOPO: fattorizzazione di Cholesky di una matrice A simmetrica

c         definita positiva nella forma A=L*L', con L matrice

c         triangolare inferiore ed L' sua trasposta.

c 

c  PARAMETRI DI INPUT:  

c 

c  - a,  vettore di reali,  elementi del triangolo inferiore di A

c                           (in formato packed)

c  - n,            intero,  ordine della matrice

c 

c  PARAMETRI DI OUTPUT:

c

c  -     l,  vettore di reali,  elementi del triangolo inferiore di L

c  - ifail,            intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna

c

c

c       |a1 b1 c1 d1 e1|       |l11  0   0   0   0 |

c       |b1 a2 b2 c2 d2|       |l21 l22  0   .   . |

c   A = |c1 b2 a3 b3 c3|   L = |l31 l32 l33  .   . |   A = L*L’

c       |d1 c2 b3 a4 b4|       |l41 l42 l43 l44  . |

c       |e1 d2 c3 b4 a5|       |l51 l52 l53 l54 l55|

c  

c

c   l = |a1 b1 a2 c1 b2 a3 .. b4 a5| triangolo inferiore di A in 

c                                    formato packed.

c    

c                   j-1

c   ljj = sqrt(ajj – (l2jk))

c                   k=1

c               j-1 

c          aij - lik ljk
c               k=1  

c   lij = ----------------

c              ljj

c

       integer n,ifail

       real a(n*(n+1)/2),l(n*(n+1)/2)

c  Controllo della dimensione      

       if (n .le. 0) then

          ifail=3

          goto 999

       endif

c  Fattorizzazione di Cholesky       

       do j=1,n

c  calcolo di ajj-somma l2jk          

          id=j*(j-1)/2+j

          l(id)=a(id)

          do k=1,j-1

             l(id)=l(id)-l(j*(j-1)/2+k)**2      

          end do

c     Controlla che la matrice sia definita positiva

c     per poter effettuare la radice quadrata di l(id)

          if (l(id) .lt. 0) then

             ifail=5

             goto 999

          endif

          l(id)=sqrt(l(id))

c     calcolo di lij          

          do i=j+1,n

             il=i*(i-1)/2+j

             l(il)=a(il)

             do k=1,j-1

                l(il)=l(il)-l(i*(i-1)/2+k)*l(j*(j-1)/2+k)

             end do

             l(il)=l(il)/l(id)    

          end do   

       end do

       ifail=0

  999  continue

      End

Risoluzione di un sistema lineare con A matrice simmetrica definita positiva con back e forward substitution

      subroutine schol(l,b,n,ifail)

c    

c     SCOPO: risoluzione di un sistema lineare, con la matrice

c            dei coefficienti simmetrica definita positiva e 

c            fattorizzata con il metodo di Cholesky: A=L*L', 

c            con L matrice triangolare inferiore e L' sua trasposta

c

c     PARAMETRI DI INPUT:

c     

c     - l,    vettore di reali,    elementi del triangolo inferiore di A

c     - b,    vettore di reali,    termine noto

c     - n,              intero,    oridine della matrice

c

c     PARAMETRI DI OUTPUT:

c

c     -     b,  vettore di reali,  soluzione del sistema

c     - ifail,            intero,  indicatore di errore

c

c     ROUTINE RICHIAMATE:

c

c     - nessuna

c

c

c     Nota implementativa - backward substitution versione colonna

c                           formato packed:

c     Nella b.s., si lavora su L'; si effettua un ciclo su j da n a 2 e

c     internamente a j, si cicla su i da 1 a j-1.

c     In realtà non si opera su L' (per operare su L', occorrerebbe

c     effettuare la trasposizione di L), ma si effettuano le computazioni

c     su L. Operando su L, trasposta di L', ciclerò su i da n a 2 e 

c     internamente a i su j da 1 a i-1

c

c     es. matrice 3x3

c                                                     j

c                                                 <-----+

c           |a11    0    0 |           |a11   a21   a31||i

c     L =  ^|a21   a22   0 |      L' = | 0    a22   a32|v

c         i||a31   a32  a33|           | 0     0    a33| 

c          +---->

c             j

c

      integer n, ifail, log

      real l(n*(n+1)/2), b(n)

c     controllo della dimensione

      if (n .le. 0) then

          ifail = 3

          goto 999

      end if

c     controlla che la matrice sia definita positiva

      do j=1,n

          k=j*(j-1)/2 +j

          if (l(k) .lt. 0) then

              ifail = 5

              goto 999

          end if

      end do

c     L*y = b

c     forward substitution versione colonne formato packed

      do j=1, n-1

          k= j+j*(j-1)/2

          b(j)= b(j)/l(k)

          do i=j+1,n

              k = j+i*(i-1)/2

              b(i)=b(i)-b(j)*l(k)

          end do

      end do

      k=n+n*(n-1)/2

      b(n)=b(n)/l(k)

c     U*x = y

c     bacward substitution, versione colonne formato packed

c     non si effettua la trasposizione della matrice triangolare L

c     ma si invertono i cicli su righe e colonne

      do i=n,2,-1

          k=i+i*(i-1)/2

          b(i)=b(i)/l(k)

          do j=1,i-1

              k=j+i*(i-1)/2

              b(j)=b(j)-b(i)*l(k)

          end do

      end do

      b(1)=b(1)/l(1)

      ifail = 0

  999  continue

       end

Soluzione di un sistema lineare sparso (Gauss-Seidell):

       Subroutine sspar(r, c, irc, n, nrc, b, tolr, resr, nmax, x0, x, ifail)

c

c  SCOPO: risoluzione di un sistema lineare sparso.

c

c  PARAMETRI DI INPUT:                                      

c                              

c  -    r,   vettore di reali,  elementi non nulli della matrice

c                               dei coefficienti del sistema

c  -    c,  vettore di interi,  indici di colonna nella matrice

c                               dei coefficienti degli elementi di r

c  -  irc,  vettore di interi,  posizioni in r del primo elemento

c                               non nullo di ciascuna riga

c  -  nrc,             intero,  numero di elementi non nulli

c  -    n,             intero,  ordine del sistema

c  -    b,   vettore di reali,  vettore dei termini noti

c  - tolr,              reale,  tolleranza relativa richiesta

c  - resr,              reale,  maggiorazione per il residuo

c                               relativo del sistema

c  - nmax,             intero,  numero massimo di valutazioni

c  -   x0,   vettore di reali,  approssimazione iniziale della 

c                               soluzione

c

c  PARAMETRI DI OUTPUT:

c

c  -     x,  vettore di reali,  soluzione del sistema

c  - ifail,            intero,  indicatore d'errore

c

c  ROUTINES RICHIAMATE:

c

c  - tolm,  funzione reale,  tolleranza minima relativa                              

c

c


c      esempio:

c                      i= 1 2 3 4 5 6 7 8 ......

c    |6 0 1 0 0|       r=(6,1,1,2,4,1,6,1,5,1,2) elementi non nulli

c    |1 2 0 0 0|

c  A=|0 0 4 1 0|       c=(1,3,1,2,3,4,4,5,2,3,5) indice di colonna dell'elemento

c    |0 0 0 6 1|

c    |0 5 1 0 2|     irc=(1,3,5,7,9,12) posizione in r del primo elemento 

c                                       non nullo di ciascuna riga, es il

c                                       primo elemento non nullo della terza 

c                                       riga è 4, che in r occupa la posizione 

c                                       5, quindi in irc al posto 3 (terza riga) 

c                                       viene memorizzato 5 (quinto elemento di c                                       r) c(n) contiene l'indice di colonna 

c                                       dell'n-esimo elemento,

c                             es per n=4, c(4)=2, il quarto elemento non nullo 

c                             della matrice (corrisponde ad r(4)=2) è sulla 

c                             seconda colonna

c

c  Gauss-Seidell:

c

c   x(0) arbitrario

c    (k+1)             i-1    (k+1)    n      (k)
c   xi    = (bi - aijxj  -   aijxj)/aii
c                j=1             j=i+1
c   i=1,..,n

c

       integer nrc, n, c(nrc), irc(n+1), nmax, ifail

       real r(nrc), tolr, resr, b(n), x0(n), x(n)

       real tolm

       external tolm

       real maxn, maxd

       logical flag  

c  Controllo delle dimensioni dei vettori

c  Il numero degli elementi non zero e la dimensione del sistema 

c  devono essere positivi

       if (nrc .le. 0  .or.  n .le. 0) then

          ifail=3

          goto 999

       endif

c  Controllo delle tolleranze relative

       if (tolr .lt. 0  .or.  tolr .ge. 1  .or.  

     %     resr .lt. 0  .or.  resr .ge. 1) then

          ifail=1

          goto 999

       endif

c  Controllo degli elementi diagonali

       do 30 i=1,n

c  irc(i) conterrà la posizione in r del primo elemento

c  non nullo della riga i-esima 

          l=irc(i)

   10     continue         

          if (c(l) .ge. i  .or.  l .eq. irc(i+1)-1) goto 20

             l=l+1

          goto 10

   20     continue

          if (c(l) .eq. i .and. r(l) .eq. 0 .or. c(l) .gt. i) then

             ifail=10 

             goto 999

          endif

   30  continue             

c  Inizializzazione 

       flag=.false.

c  k conta il numero di iterazioni eseguite

       k=0  

   40  continue

c  controllo del raggiungimento del numero massimo di iterazioni

         if (k .eq. nmax) then

            ifail=9

            goto 999

         endif                                                  

c  Fattorizzazione (Gauss-Seidell)

         l=0

         do 60 i=1,n

            x(i)=b(i)

c  num contiene il numero di elementi appartenenti alla i-esima riga, 

c  irc(i+1) contiene la posizione in r del primo elemento della 

c  (i+1)-esima riga, mentre irc(i), contiene la posizione in r 

c  del primo elemento della i-esima riga.

c  num => sulla i-esima riga sono presenti num elementi non nulli

            num=irc(i+1)-irc(i)             

c  ciclo per gli elementi della i-esima riga

            do 50 j=1,num

               l=l+1

c  sum_j=1^i-1

c  se sto esaminando colonne con indice minore ad i

               if (c(l) .lt. i) then                 

                  x(i)=x(i)-r(l)*x(c(l))

c  aii = r(l)

c  altrimenti se sto esaminando la colonna di indice i

               elseif (c(l) .eq. i) then

                  d=r(l)                              

c  sum_j=i+1^n

c  se sto esaminado colonne di indice maggiore di i

               elseif (c(l) .gt. i) then

                  x(i)=x(i)-r(l)*x0(c(l))

               endif

   50       continue            

            x(i)=x(i)/d

60 continue            

         k=k+1

c    Calcolo della norma infinito ||x-x0|| e ||x||

         maxn=abs(x(1)-x0(1))

         maxd=abs(x(1))

         do 70 i=2,n

            temp=abs(x(i)-x0(i))

            if (temp .gt. maxn) maxn=temp

            temp=abs(x(i))

            if (temp .gt. maxd) maxd=temp

   70    continue

c    Test sulla convergenza

         if (maxn .le. tolr*maxd+tolm(maxd)) then

            flag=.true.

         else

c       Calcolo della norma infinito ||b-A*x|| e ||b||

            maxn=abs(b(1)-r(1)*(x(1)-x0(1)))

            maxd=abs(b(1))

            do 100 i=2,n  

c          Ricerca dell'indice dell'elemento diagonale   

               l=irc(i)

   80          continue

               if (c(l) .eq. i) goto 90

                  l=l+1

               goto 80

   90          continue         

               temp=abs(b(i)-r(l)*(x(i)-x0(i)))

               if (temp .gt. maxn) maxn=temp

               temp=abs(b(i))

               if (temp .gt. maxd) maxd=temp

  100       continue

c       Test sulla bontà del risultato

            if (maxn .le. resr*maxd+tolm(maxd)) flag=.true.

         endif

c    Copia di x in x0

         do 110 i=1,n

            x0(i)=x(i)

  110    continue

       if (.not. flag) goto 40

       ifail=0

  999  continue

      End

Calcolo dei coefficienti del polinomio interpolante di Newton:

Subroutine conwt(x, y, n, a, ifail)

c

c  SCOPO: calcolo dei coefficienti del polinomio interpolante 

c         nella forma di Newton.

c        

c  PARAMETRI DI INPUT:  

c 

c  - x,  vettore di reali,  ascisse dei punti di interpolazione

c  - y,  vettore di reali,  ordinate dei punti di interpolazione

c  - n,            intero,  numero di punti di interpolazione

c 

c  PARAMETRI DI OUTPUT:

c

c  -     a, vettore di reali,  coefficienti del polinomio

c  - ifail,           intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna

c

c   Algoritmo differenze divise

c

c      |k=0     1       2     . . .      n

c   ---+-------------------------------------

c    x0| y0

c      |       y01

c    x1| y1            y012

c      |       y12

c    x2| y2            y123           y012..n   

c    . | .     y23      . 

c    . | .      .       .

c      | .      .      yn-2n-1n

c      | .     yn-1n

c    xn| yn

c

c

c                    y[xi+1,..,xi+k]-y[xi, ..,xi+k-1]

c   y[xi, ..,xi+k]= ----------------------------------

c                             xi+k – xi

c

       integer n,ifail

       real x(n),y(n),a(n)

c     Controllo del numero dei punti

       if (n .le. 0  .or.  n .gt. 9) then

          ifail=11

          goto 999

       endif

c     Controlla che tutte le ascisse siano distinte   

       do i=1,n

          do j=i+1,n

             if (x(i)-x(j) .eq. 0) then

                ifail=12

                goto 999

             endif

          end do

       end do

c
Differenze divise

c     Inizializzazione del vettore a 

       do i=1,n

          a(i)=y(i)

       end do

c     Calcolo delle differenze divise   

       do k=1,n-1

          do i=1,n-k

             a(i)=(a(i+1)-a(i))/(x(i+k)-x(i))

          end do

       end do

       ifail=0

  999  continue

      End

Calcolo del valore di un polinomio in un punto:

Subroutine nwtn(p, a, n, x, m, y, ifail)

c

c  SCOPO:Calcolo del valore di un polinomio in un punto.

c        

c  PARAMETRI DI INPUT:  

c 

c  - p,  vettore di reali,  ascisse dei punti di interpolazione

c  - a,  vettore di reali,  coefficienti del polinomio interpolante

c  - n,            intero,  numero di punti di interpolazione

c  - x,  vettore di reali,  punti in cui valutare il polinomio

c  - m,            intero,  numero di punti in cui valutare il polinomio

c 

c  PARAMETRI DI OUTPUT:

c

c  -     y, vettore di reali, valori calcolati

c  - ifail,           intero, indicatore d'errore

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna

c

c  I coefficienti del polinomio di interpolazione sono

c  stati calcolati con la subroutine conwt. 

c

       integer n,m,ifail

       real p(n),a(n),x(m),y(m)

c      Controllo sul numero di punti

       if (n .le. 0  .or.  m .le. 0) then

           ifail=11

           goto 999

        endif

       do 10 i=1,m

          y(i)=a(1)

          do 10 j=2,n

             y(i)=y(i)*(x(i)-p(j))+a(j)

   10     continue

   20  continue

  999  continue         

       End

Calcolo dei coefficienti di un polinomio dei minimi quadrati di 1° grado

Subroutine pmq1(x, y, n, c, r, ifail)

c

c  SCOPO: determinazione dei coefficienti del polinomio dei minimi

c         quadrati di 1° grado.

c 

c  PARAMETRI DI INPUT:  

c 

c  - x,  vettore di reali,  ascisse dei punti

c  - y,  vettore di reali,  ordinate dei punti

c  - n,            intero,  numero di punti

c 

c  PARAMETRI DI OUTPUT:

c

c  -     c,  vettore di reali,  coefficienti del polinomio

c  -     r,  vettore di reali,  vettore dei residui

c  - ifail,            intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - chol,fsbpk,bsbpk

c  

c  --------- 

c  n*c1 + (sum_i=1^n xi)c2 = (sum_i=1^n yi)

c   

c  (sum_i=1^n xi)*c1 + (sum_i=1^n xi*xi)*c2 = (sum_i=1^n yi*xi)

c  

c  La soluzione del sistema in alto da i coefficienti c1 e c2

c  ---------

c

c  B = AtA -=-=-  d=Aty

c  B = LLt -=-=-  Lw=d  -=-=- Lc=w

c

c  Dati n punti (x0,y0), (x1, y1), .., (xn,yn), la matrice A è

c     |1 x0|               
c     |1 x1|               |n   |

c  A =|1 x2|  => B = AAt = |     |  simmetrica definita positiva

c     |. . |               |   |
c     |1 xn|               
c                    |y0|   

c      |1  1  ..  1| |..|   |d1|

c  d = |x0 x1 .. xn| |..| = |..| vettore termini noti

c                    |yn|   |dn|

c 

c           |n    xi|

c  B = AAt =|         |  B deve essere fattorizzata con Chol, 

c           |xi   x2i|  il vettore da passare è |n xi  x2i|  

c

       external chol,fsbpk,bsbpk

       integer n, ifail          

       real    x(n), y(n), c(2), r(n)

       real    b(3), d(2), l(3)

c   Variabili di appoggio:

c   b - vettore simulante la matrice B dei coefficienti

c   d – vettore dei termini noti

c   l – vettore simulante la matrice fattorizzata

       ifail=0

c  Controllo del numero di punti

       if (n .le. 0) then

          ifail=11

          goto 999

       endif

       if (n .eq. 1) then

          ifail=13

          goto 999

       endif

       if (n .eq. 2) ifail=14

c  Controllo delle ascisse

       do 2 i=2,n

          if (x(1) .ne. x(i)) goto 2

    1 continue          

       ifail=13

       goto 999

    2 continue

c  Calcolo dei coefficienti e dei termini noti

c  del sistema delle equazioni normali

       b(1)=n

       b(2)=0

       do 10 i=1,n

          b(2)=b(2)+x(i)

   10  continue

       b(3)=0

       do 20 i=1,n

          b(3)=b(3)+x(i)**2

   20  continue  

       c(1)=0

       do 30 i=1,n

          c(1)=c(1)+y(i)

   30  continue

       c(2)=0 

       do 40 i=1,n

          c(2)=c(2)+x(i)*y(i)

   40  continue   

c  Fattorizzazione di Cholesky della matrice dei coefficienti

c  Calcolo di B = AtA

       call chol(b,l,2,ifail)

c  Risoluzione del sistema  

c  Calcolo di Lw=d --- Ltc=w

       call fsbpk(l,c,2,ifail)

       call bsbpk(l,c,2,ifail)

c  Calcolo dei residui

       do 50 i=1,n

          r(i)=y(i)-(c(1)+c(2)*x(i))

   50  continue                   

  999  continue

       End

Calcolo dei coefficienti di un polinomio dei minimi quadrati di 2° grado
Subroutine pmq2(x, y, n, c, r, ifail)

c

c  SCOPO: calcolo dei coefficienti del polinomio dei minimi quadrati

c         di secondo grado.

c 

c  PARAMETRI DI INPUT:  

c 

c  - x,  vettore di reali,  ascisse dei punti

c  - y,  vettore di reali,  ordinate dei punti

c  - n,            intero,  numero dei punti

c 

c  PARAMETRI DI OUTPUT:

c

c  -     c,  vettore di reali,  coefficienti

c  -     r,  vettore di reali,  vettore dei residui

c  - ifail,            intero,  indicatore di errore

c   

c  ROUTINES RICHIAMATE:

c

c  - chol,fsbpk,bsbpk

c  

c  --------- 

c  n*c1 + (sum_i=1^n xi)c2 + (sum_i=1^n xi^2)c3= (sum_i=1^n yi)
c   

c  (sum_i=1^n xi)*c1 + (sum_i=1^n xi^2)*c2 +(sum_i=1^n xi^3)c3 = (sum_i=1^n yi*xi)

c
c  (sum_i=1^n xi^2)*c1 + (sum_i=1^n xi^3)*c2 +(sum_i=1^n xi^4)c3 = (sum_i=1^n yi*xi^2)

c
c  La soluzione del sistema in alto da i coefficienti c1 c2 e c3

c  ---------

c

c  B = AtA 

c  d=Aty

c  B = LLt 

c  Lw=d  

c  Lc=w

c      |n      |

c  B = |      |  simmetrica definita positiva

c      |      |
c      

c      |d1|

c  d = |..| vettore termini noti

c      |dn|

c 

c      |n   xi  xi2|

c  B = |xi xi2 xi3|  B deve essere fattorizzata con Chol, 

c      |xi2 xi3 xi4|  il vettore da passare è |n xi  x2i xi2 xi3 xi4|  

c

       external chol, fsbpk, bsbpk

       integer ifail          

       real    x(n), y(n), c(3), r(n)

       real    b(6),l(6)

c  Variabili di appoggio

c   b - vettore simulante la matrice B dei coefficienti

c   l – vettore simulante la matrice fattorizzata

       ifail=0

c  Controllo del numero di punti

       if (n .le. 0) then

          ifail=11

          goto 999

       endif 

       if (n .le. 2) then

          ifail=13

          goto 999

       endif      

       if (n .eq. 3) ifail=14

c  Controllo delle ascisse

       do 1 i=2,n

          if (x(1) .ne. x(i)) goto 2

    1  continue

       ifail=13

       goto 999

    2  continue

c  Inizializzazione del vettore r(i)

       do 10 i=1,n

          r(i)=x(i)

   10  continue

c  Calcolo dei coefficienti e dei termini noti del sistema

c  di equazioni normali

       b(1)=n

       c(1)=0

       do 20 i=1,n

          c(1)=c(1)+y(i)

   20  continue

       b(2)=0

       c(2)=0

       do 30 i=1,n

          b(2)=b(2)+r(i)

          c(2)=c(2)+r(i)*y(i)

   30  continue

       b(3)=0

       do 40 i=1,n

          r(i)=r(i)**2

          b(3)=b(3)+r(i)

          c(3)=c(3)+r(i)*y(i)

   40  continue     

       b(4)=b(3)

       do 60 k=5,6

          b(k)=0

          do 50 i=1,n

             r(i)=r(i)*x(i)

             b(k)=b(k)+r(i)

   50     continue        

   60  continue

c  Fattorizzazione di Cholesky della matrice dei coefficienti b=l*l'

       call chol(b,l,3,ifail)

c  Risoluzione del sistema l*w=d

       call fsbpk(l,c,3,ifail)

c  Trasposta della matrice l

       swap=l(3)

       l(3)=l(4)

       l(4)=swap

c  Risoluzione del sistema l'*c=w

       call bsbpk(l,c,3,ifail)

c  Calcolo dei residui

       do 70 i=1,n

          r(i)=y(i)-(c(1)+c(2)*x(i)+c(3)*x(i)**2)

   70  continue       

  999  continue

      End

Calcolo dei coefficienti di spline cubica:

Subroutine cospl(x, y, n, a, b, c, d, ifail)

c

c  SCOPO: calcolo dei coefficienti della spline cubica naturale

c         interpolante n punti assegnati. 

c

c  PARAMETRI DI INPUT:

c

c  - x,  vettore di reali,  ascisse dei punti di interpolazione

c  - y,  vettore di reali,  ordinate dei punti di interpolazione

c  - n,            intero,  numero dei punti di interpolazione

c

c  PARAMETRI DI OUTPUT:

c

c  -     a,  vettore di reali,  coefficienti dei termini di grado

c                               3 degli n-1 polinomi interpolanti

c  -     b,  vettore di reali,  coefficienti dei termini di grado

c                               2 degli n-1 polinomi interpolanti

c  -     c,  vettore di reali,  coefficienti dei termini di grado

c                               1 degli n-1 polinomi interpolanti

c  -     d,  vettore di reali,  coefficienti dei termini di grado

c                               0 degli n-1 polinomi interpolanti

c  - ifail,            intero,  indicatore di errore

c

c  ROUTINE RICHIAMATE:

c

c  -  lut,  procedura,  fattorizzazione LU per matrici tridiagonali

c  - slut,  procedura,  risoluzione di un sistema tridiagonale

c

c   

c   s3(x) = c1,j + c2,j(x-xj) + c3,j(x-xj)2 + c4,j(x-xj)3
c

c   La matrice simmetrica tridiagonale dei coefficienti di grado 3 2 1 è:

c 

c       |2(h2+h3)    h3       0      .    .      .      |

c       |    h3   2(h3+h4)    h4     0    .      .      |

c   A = |    0       h4    2(h4+h5)  h5   0      .      | 

c       |    .       0         .     .    .      .      |

c       |    .       .         .     .    .     hn-1    | 

c       |    0       .         .     .   hn-1 2(hn-1+hn)| 

c

c   hj+1 = xj+1-xj
c

c   risolto il sistema e ottenuti i valori delle incognite Mj
c   si possono calcolare i coefficienti della spline naturale

c

c   Mi = di
c   c1,j = yj


c2,j = (yj+1-yj)/hj+1 – (2Mj + Mj+1)hj+j/6

c   c3,j =Mj/2

c4,j = (Mj+1-Mj)/6hj+1
c

c

       external lut, slut

       integer n, ifail

       real    x(n), y(n), a(n-1), b(n-1), c(n-1), d(n-1)

       logical ord

c  Controllo del numero di punti

        if (n .le. 2) then

           ifail=11

           goto 999

        endif

c  Ordinamento delle ascisse

       ord = .false.

       k = n

       do while (.not. ord)

c   10  continue

          ord=.true.

          do i=1,k-1    

             if (x(i) .gt. x(i+1)) then

                t=x(i)

                x(i)=x(i+1)

                x(i+1)=t

                ord=.false.

             endif  

          end do

          k=k-1   

       end do

c      if (.not. ord) goto 10

c  Controllo della non coincidenza delle ascisse

       do i=1,n-1

          if (x(i) .eq. x(i+1)) then

             ifail=12

             goto 999

          endif

       end do

c  Calcolo dei coefficienti e dei termini noti del sistema

c  avente come soluzione le derivate seconde della spline

c  nelle ascisse interne dei punti di interpolazione

       do i=1,n-1

          a(i)=x(i+1)-x(i)

       end do

       do i=2,n-2

          b(i)=a(i)

       end do

       do i=1,n-2

          c(i)=2*(a(i)+a(i+1))

       end do

       do i=1,n-2

          d(i+1)=6*((y(i)-y(i+1))/a(i) - (y(i+1)-y(i+2))/a(i+1))

       end do

c  Fattorizzazione LU della matrice dei coefficienti

       call lut(c,b(2),a(2),n-2,ifail)

c  Risoluzione del sistema tridiagonale

       call slut(c,b(2),a(2),d(2),n-2,ifail)

c  il vettore d contiene la soluzione della matrice

c  tridiagonale, cioè il valore delle incognite Mj

c  utilizzate per calcolare i coefficienti della spline

c  Derivata seconda nella prima ascissa

       d(1)=0       

c  Calcolo dei coefficienti in funzione delle derivate seconde

       do i=1,n-2

          c(i)=(y(i+1)-y(i))/a(i) - (d(i+1)+2*d(i))/6*a(i)

       end do

       c(n-1)=(y(n)-y(n-1))/a(n-1) - 2*d(n-1)/6*a(n-1)

       do i=1,n-1

          b(i)=d(i)/2

       end do

       do i=1,n-2

          a(i)=(d(i+1)-d(i))/6/a(i)

       end do

          a(n-1)=-d(n-1)/6/a(n-1)

       do i=1,n-1

          d(i)=y(i)

       end do

       ifail=0

  999  continue

      End

Spline cubica:

Subroutine splin(p, a, b, c, d, n, m, x, y, ifail)

c                

c  SCOPO: valutazione della spline cubica naturale di assegnati

c         coefficienti.

c

c  PARAMETRI DI INPUT:

c

c  - p,  vettore di reali,  ascisse dei punti di interpolazione

c  - a,  vettore di reali,  coefficienti dei termini di grado 3

c  - b,  vettore di reali,  coefficienti dei termini di grado 2

c  - c,  vettore di reali,  coefficienti dei termini di grado 1

c  - d,  vettore di reali,  coefficienti dei termini di grado 0

c  - n,            intero,  numero di punti di interpolazione

c  - m,            intero,  numero di punti in cui valutare la

c                           spline

c  - x,  vettore di reali,  punti in cui valutare la spline

c

c  PARAMETRI DI OUTPUT:

c

c  -     y,  vettore di reali,  valori della spline nei punti di x

c  - ifail,            intero,  indicatore d'errore

c

c  ROUTINE RICHIAMATE:

c

c  - nessuna

c

       integer n, ifail

       real p(n), a(n-1), b(n-1), c(n-1), d(n-1), x(m), y(m)

       logical ord               

       integer primo, ultimo

c  Controllo del numero di punti

        if (n .lt. 3) then

           ifail=11

           goto 999

        endif

c  Ordinamento delle ascisse

       k=n

   10  continue

          ord=.true.

          do 20 i=1,k-1   

             if (p(i) .gt. p(i+1)) then

                t=p(i)

                p(i)=p(i+1)

                p(i+1)=t

                ord=.false.

             endif  

   20     continue

          k=k-1   

       if (.not. ord) goto 10

c  Controllo dell'assenza di ascisse uguali

       do 25 i=1,n-1

          if (x(i) .eq. x(i+1)) then

             ifail=12

             goto 999

          endif

   25  continue

c  Controllo dell'appartenenza dei punti di x all'intervallo [p(1),p(n)]

       do 30 i=1,m

          if (x(i) .lt. p(1)  .or.  x(i) .gt. p(n)) then

             ifail=15

             goto 999

          endif

   30  continue

       do 60 j=1,m

c     Ricerca binaria di x(j) in p

          primo=1

          ultimo=n     

   40     continue

          i=(primo+ultimo)/2

          if (x(j) .ge. p(i)  .and.  x(j) .le. p(i+1)) goto 50

             if (x(j) .lt. p(i)) ultimo=i

             if (x(j) .gt. p(i+1)) primo=i+1

          goto 40

   50     continue

c     Valutazione della spline in x(j)

          h=x(j)-p(i)

          y(j)=((a(i)*h+b(i))*h+c(i))*h+d(i)

   60  continue

       ifail=0

  999  continue

      End

Fast Fourier Transform (radix 2):

Subroutine fft2(flag, f, n, ifail)

c

c  SCOPO:calcolo della trasformata discreta di Fourier "radix 2"

c        o della sua inversa.

c        

c  PARAMETRI DI INPUT:  

c 

c  - flag,              booleano,  indicatore DFT/IDFT

c  -    f,  vettore di complessi,  vettore da trasformare

c  -    n,                intero,  ordine della trasformata

c   

c  PARAMETRI DI OUTPUT:

c

c  -     f,  vettore di complessi,  componenti della trasformata

c  - ifail,                intero,  indicatore d'errore

c   

c  ROUTINES RICHIAMATE:

c

c  - btrev

c        

c    

       integer btrev

       external btrev

       real twopi

       parameter (twopi=6.283185308)

c   flag = .true.  calcola la fft, 

c   flag = .false. calcola la idft

       logical flag

       integer n,ifail

       complex f(0:n-1)

       integer d,nesp,ind1,ind2,s,z,st

       real esp

       complex e,t

c  Controllo dell'ordine

       if (n .le. 0) then

          ifail=20

          goto 999

       endif

c  Controllo che n sia potenza di due

       k=n

       m=0 

       do while(mod(k,2) .eq. 0)

          m=m+1

          k=k/2

       end do

       if (k .ne. 1) then

          ifail=21

          goto 999

       endif

c  DFT/IDFT

       if (flag) then 

          st=-1 

       else 

          st=1

       endif

       d=n

c  Numero di esponenziali diversi al primo passo

       nesp=1

       do k=1,m       

c    Distanza tra i nodi duali al passo k  

          d=d/2

          s=0

          do i=1,nesp        

c      Calcolo dell'esponenziale     

             z=s

c   Nota:

c   per compilare la subroutine in linux, occorre cambiare la funzione

c   di libreria isha, (presente nella libreria del fortran Microsoft)

c   con l’equivalente funzione di libreria del fortran Gnu, ishift. 

c   Entrambe effettuano lo shift aritmetico del primo valore, della 

c   quantità definita dal secondo. Se il secondo valore è maggiore di

c   zero lo shift è effettuato a sinistra, se è minore di zero, lo

c   shift è effettuato a destra.

             z=isha(z,-(m-k))

c            z=ishift(z,-(m-k))  ## Linux

             z=btrev(z,k)

             esp=st*twopi*z/2**k

             e=exp(cmplx(0,esp))

             do l=0,d-1

c        Posizione del primo elemento della coppia duale

                ind1=l+s

c        Posizione del secondo elemento della coppia duale

                ind2=ind1+d

c        Calcolo di una DFT di ordine 2        

                t=e*f(ind2)

                f(ind2)=f(ind1)-t

                f(ind1)=f(ind1)+t 

              end do

             s=s+2*d

             end do 

c      Numero di esponenziali diversi al passo k+1    

          nesp=nesp*2

          end do

c  Ordinamento del vettore f

       do k=1,n-2

          i=btrev(k,m)

          if (i .gt. k) then

             t=f(k)

             f(k)=f(i)

             f(i)=t

          endif

       end do

c  se si esegue la IDFT il vettore f viene diviso per n      

       if (.not. flag) then

          do i=0,n-1

             f(i)=f(i)/n

          end do

       endif

       ifail=0

  999  continue

      End

Bit reversal:

integer function btrev(z,k)

c  

c  SCOPO: operazione di bit-reversal degli ultimi k bit della 

c         rappresentazione binaria di un intero

c

c  PARAMETRI DI INPUT:

c

c  -     z,  intero,  operando

c  -     k,  intero,  numero di bit da invertire

c

c  PARAMETRI DI OUTPUT:

c

c  - z,  intero,  risultato

c                       

c  ROUTINES RICHIAMATE:

c

c  - nessuna

c

c

c                      k               i    

c        |-------------| |-------------|

c    z1= 0 1 0 0 1 0 1 1 0 1 1 0 0 1 1 0

c

c

       integer z,k

       integer z1

       z1=z

       do i=1,k/2

          if (btest(z1,i-1) .and. (.not. btest(z1,k-i))) then

             z1=ibclr(z1,i-1)

             z1=ibset(z1,k-i)

          elseif ((.not. btest(z1,i-1)) .and. btest(z1,k-i)) then

             z1=ibset(z1,i-1)

             z1=ibclr(z1,k-i)

          endif

       end do

       btrev=z1

      End

Quadratura di una funzione reale:

Subroutine quad(f, a, b, tol, hmin, nmax, q, e, nval, ifail)

c

c  SCOPO: calcolo dell'integrale definito di una funzione

c         reale di una variabile reale.

c 

c  PARAMETRI DI INPUT:  

c 

c  -      f,  funzione reale,  funzione integranda

c  -      a,           reale,  estremo sinistro 

c                              dell'intervallo d'integrazione

c  -      b,           reale,  estremo destro 

c                              dell'intervallo d'integrazione

c  -    tol,           reale,  tolleranza di calcolo richiesta

c  -   hmin,           reale,  passo minimo d'integrazione

c  -   nmax,          intero,  numero massimo di valutazioni

c

c  PARAMETRI DI OUTPUT:

c

c  -     q,   reale,  stima dell'integrale definito di f    

c  -     e,   reale,  stima dell'errore di calcolo

c  -  nval,  intero,  numero minimo di valutazioni occorse

c  - ifail,  intero,  indicatore d'errore

c   

c  ROUTINES RICHIAMATE:

c

c  -  eps,  funzione reale,  epsilon macchina

c  - simp,  funzione reale,  calcolo dell'integrale mediante

c                            la formula di Cavalieri-Simpson

c  - tolm,  funzione reale,  tolleranza minima relativa

c

       real simp, eps, tolm

       external simp, eps, tolm

       external f

       integer dim

       parameter (dim=25)                  

       real    a, b, tol, hmin, q, e

       integer nmax, nval, ifail

       real    h, stack(dim,3), c, d, pm, s1, s2, eloc

       integer n

c  Controllo della tolleranza assoluta

       if (tol .lt. 0) then

          ifail=2

          goto 999

       endif

c  Calcolo dell'ampiezza dell'intervallo di integrazione

       h=abs(b-a)

c  Inizializzazione delle variabili di uscita

       q=0

       e=0

       nval=0

       ifail=0

c  Controllo del numero di valutazioni massime eseguibili

       if (nmax .lt. 5) then

          ifail=16

          goto 999

       endif

c  Controllo degli estremi dell'intervallo

       if (b-a .eq. 0) goto 999

c  Controllo del passo minimo d'integrazione

       if (hmin .le. 0 .or. hmin .ge. h) then

          ifail=17

          goto 999

       endif

       if (hmin .lt. 8*eps()*h) hmin=8*eps()*h

c  Inizializzazione dello stack

c  Lo stack contiene, all’inizializzazione, gli estremi

c  di integrazione e la prima stima dell’integrale.

c  Se a > b l'integrale viene calcolato tra b e a

       stack(1,1)=min(a,b)

       stack(1,2)=max(a,b)

       stack(1,3)=simp(f,stack(1,1),stack(1,2))

       n=1

       nval=nval+3

c  Ciclo a condizione: termina se lo stack è vuoto o si 

c  raggiunge il numero massimo di valutazioni 

   10  continue

c  Prelievo dallo stack degli estremi dell'intervallo corrente

c  d'integrazione e dell'integrale relativo ad esso

       c=stack(n,1)

       d=stack(n,2)

       s1=stack(n,3)

c  decremento lo stack 

       n=n-1

c  Calcolo del punto medio 

       if (c*d .lt. 0) then

          pm=(c+d)/2      

       else

          pm=c+(d-c)/2

       endif

c  Calcolo della somma dei due integrali relativi alle

c  due metà dell'intervallo corrente d'integrazione

       s2=simp(f,c,pm)+simp(f,pm,d)

c  Calcolo dell'errore locale       

       eloc=abs(s1-s2)/15

c  Controllo dell'errore locale e del passo d'integrazione

       if (eloc .lt. tol*(d-c)/h+tolm(s1) .or. d-c .le. hmin) then

          q=q+s2

          e=e+eloc

       else                           

c  Controllo della lunghezza dello stack  

          if (n .gt. dim-2) then

             ifail=18

             n=0

          else   

             n=n+1

             stack(n,1)=pm

             stack(n,2)=d

             stack(n,3)=simp(f,pm,d)

             n=n+1

             stack(n,1)=c

             stack(n,2)=pm       

             stack(n,3)=simp(f,c,pm)

          endif

       endif

c  Calcolo del numero di valutazioni 

       nval=nval+2

       if (n .gt. 0 .and. nval .lt. nmax) goto 10

c  Valore dell'integrale

c  Se a < b si moltiplica per -1

c  l'integrale definito tra b ed a

       q=sign(1.0,b-a)*q

c  Controllo dello stack                               

       if (n .gt. 0) ifail=19

  999  continue

      End

Integrale approssimato di una funzione (Cavalieri-Simpson)

real function simp(f, a, b)

c

c  SCOPO: Calcolo approssimato dell'integrale definito di una

c         funzione mediante la formula di Cavalieri-Simpson.

c 

c  PARAMETRI DI INPUT:  

c 

c  - f,  funzione reale,  funzione integranda

c  - a,           reale,  estremo sinistro 

c                         dell'intervallo d'integrazione

c  - b,           reale,  estremo destro

c                         dell'intervallo d'integrazione

c 

c  PARAMETRI DI OUTPUT:

c

c  - simp,  reale,  valore dell'integrale    

c   

c  ROUTINES RICHIAMATE:

c

c  - nessuna

c

c  Modulo di integrazione 

c

c

       external f

       real a,b

       if (b-a .eq. 0) then

          simp=0

       else

c   Calcolo del punto medio 

          if (a*b .lt. 0) then

             pm=(a+b)/2      

          else

             pm=a+(b-a)/2

          endif

c   Calcolo dell'integrale          

          simp=(b-a)/6*(f(a)+4*f(pm)+f(b))

       endif

      End

Calcolo dello zero di una funzione (metodo delle secanti)

      Subroutine sec(f,x1,x2,tolx,tolf,nmax,z,fz,nval,ifail)

c

c  SCOPO: risoluzione dell'equazione f(x)=0 

c         calcolo dello zero di f

c 

c  PARAMETRI DI INPUT:  

c 

c  -      f,  funzione reale,  funzione di cui calcolare lo zero

c  -     x1,           reale,  primo valore di partenza 

c  -     x2,           reale,  secondo valore di partenza 

c  -   tolx,           reale,  tolleranza relativa per lo zero

c  -   tolf,           reale,  tolleranza assoluta per la funzione

c  -   nmax,          intero,  numero massimo di valutazioni

c

c  PARAMETRI DI OUTPUT:

c

c  -     z,   reale,  zero della funzione (=soluzione calcolata)    

c  -    fz,   reale,  valore della funzione (=residuo)

c  -  nval,  intero,  numero di valutazioni occorse

c  - ifail,  intero,  indicatore d'errore

c   

c  ROUTINES RICHIAMATE:

c

c  - tolm,  funzione reale,  tolleranza minima relativa

c

       real eps,tolm

       external eps,tolm

       real f

       external f

       real    x1, x2, tolx, tolf, z, fz

       integer nmax, nval, ifail

       real    s, p, q, xk, xkm1, xkm2, dx

c  xkm1 – x al passo k meno uno

c  xkm2 – x al passo k meno due


c  dx   - delta x

c  Controllo della tolleranza relativa

       if (tolx .lt. 0  .or.  tolx .ge. 1) then

          ifail=1

          goto 999

       endif

c  Controllo della tolleranza assoluta

       if (tolf .lt. 0) then

          ifail=2

          goto 999

       endif

c  Controllo del numero di valutazioni di funzione

       if (nmax .lt. 2) then

          ifail=16

          goto 999

       endif                     

c  Controllo 

       if (x1 .eq. x2) then

          ifail = 22

          goto 999

       endif

c  Inizializzazioni                

       e=eps()

       xkm1 = x1

       xkm2 = x2

       nval = 2

       ifail = 0

c  Ciclo a condizione: termina se si è raggiunta una delle

c  due tolleranze o il numero massimo di valutazioni       

   10  continue

c     Controllo del numero di valutazioni          

          if (nval .ge. nmax) then

             ifail=19

             goto 20

          endif

c   calcolo della secante per i punti 

c   (xkm1, f(xkm1)) (xkm2, f(xkm2))

c   Per evitare problemi di overflow, Il punto xk 

c   si ottiene iterativamente come segue:

          s = f(xkm1) / f(xkm2)

          p = (xkm2 - xkm1) * s

          q = 1 - s

          xk = xkm1 - p/q

          nval=nval+1

c     controllo sul raggiungimento delle

c     condizioni di uscita dal ciclo

          dx= xk - xkm1

          fz = f(xk)

          if ((abs(fz) .le. tolf+e) .or. 

     %         (abs(dx) .le. tolx*abs(xkm1)+tolm(xkm1))) goto 20

          xkm2 = xkm1

          xkm1 = xk

          goto 10

   20 continue    

c     Soluzione calcolata   

       z = xk

c     Residuo

       fz = f(xk)

  999  continue       

      End          

Calcolo dello zero di una funzione (algoritmo di Dekker-Brent):

Subroutine fzero(f, a, b, tolx, tolf, nmax, z, fz, nval, ifail)

c

c  SCOPO: risoluzione dell'equazione f(x)=0 (ovvero calcolo

c         dello zero di f) in un intervallo compatto negli

c         estremi del quale la f, funzione reale di una

c         variabile reale, assume valori di segno opposto.

c 

c  PARAMETRI DI INPUT:  

c 

c  -      f,  funzione reale,  funzione di cui calcolare lo zero

c  -      a,           reale,  estremo sinistro dell'intervallo

c                              contenente lo zero

c  -      b,           reale,  estremo destro dell'intervallo

c                              contenente lo zero

c  -   tolx,           reale,  tolleranza relativa per lo zero

c  -   tolf,           reale,  tolleranza assoluta per la funzione

c  -   nmax,          intero,  numero massimo di valutazioni

c

c  PARAMETRI DI OUTPUT:

c

c  -     z,   reale,  zero della funzione (=soluzione calcolata)    

c  -    fz,   reale,  valore della funzione (=residuo)

c  -  nval,  intero,  numero di valutazioni occorse

c  - ifail,  intero,  indicatore d'errore

c   

c  ROUTINES RICHIAMATE:

c

c  - eps,   funzione reale,  epsilon macchina

c  - tolm,  funzione reale,  tolleranza minima relativa

c

       real eps, tolm

       external eps, tolm

       real f

       external f

       real a, b, tolx, tolf, z, fz

       integer nmax, nval, ifail

c  Controllo della tolleranza relativa

       if (tolx .lt. 0  .or.  tolx .ge. 1) then

          ifail=1

          goto 999

       endif

c  Controllo della tolleranza assoluta

       if (tolf .lt. 0) then

          ifail=2

          goto 999

       endif

c  Controllo del numero di valutazioni di funzione

       if (nmax .lt. 2) then

          ifail=16

          goto 999

       endif

c  Controllo del segno della funzione negli estremi

       if (f(a)*f(b) .ge. 0) then

          ifail=22

          goto 999

       endif

c  Inizializzazioni                

       e=eps()

       x0=a

       x1=b

       y1=x0

       y0=y1*(1+2*e)

       ym1=y0

       f0=f(x0)

       f1=f(x1)     

       nval=2

       ifail=0

c  Ciclo a condizione: termina se si è raggiunta una delle

c  due tolleranze o il numero massimo di valutazioni       

   10  continue

       xy=x1-y1

       if (abs(xy) .le. tolx*abs(x1)+tolm(x1)  

     %    .or.  abs(f1) .le. tolf+e)           goto 20

c     Controllo del numero di valutazioni          

          if (nval .ge. nmax) then

             ifail=19

             goto 20

          endif

c     Controlla che un estremo non si ripeta troppe volte

          if (y1 .ne. ym1) then

             d=(f1/f0)*(x0-x1)/(1-f1/f0)

c        Controlla che la secante non esca dall'intervallo [x1,y1]                          

             if (d*xy .lt. 0  .or.  abs(d) .gt. abs(xy)) d=0.5*xy

          else

             d=0.5*xy

          endif

          x0=x1

          f0=f1

          x1=x1-d

          f1=f(x1)

          nval=nval+1

          ym1=y0

          y0=y1

          if (f0*f1 .lt. 0) y1=x0

       goto 10

   20  continue    

c  Soluzione calcolata   

       z=x1

c  Residuo

       fz=f1

  999  continue       

      End

Equazione differenziale ordinaria - RKF45

Subroutine ode45 (f, a, b, y0, m, tol, n, x, y, ifail)

c

c  SCOPO: risoluzione di un'equazione differenziale ordinaria

c         in un intervallo compatto.

c 

c  PARAMETRI DI INPUT:  

c  

c  -    f,  funzione reale,  termine noto dell'equazione

c  -    a,           reale,  estremo sinistro dell'intervallo 

c  -    b,           reale,  estremo destro dell'intervallo

c  -   y0,           reale,  valore iniziale in a

c  -    m,          intero,  numero iniziale di intervalli

c  -  tol,           reale,  tolleranza richiesta

c  -    n,          intero,  numero massimo di valori da calcolare

c 

c  PARAMETRI DI OUTPUT:

c

c  -      x,  vettore di reali,  punti di griglia

c  -      y,  vettore di reali,  valori della soluzione in x

c  -  ifail,            intero,  indicatore d'errore

c

c  ROUTINES RICHIAMATE:

c

c  -  tolm,  funzione reale,  tolleranza minima relativa

c

       real tolm

       external tolm

       integer m, n, ifail

       real    a, b, y0, tol, x(n), y(n)

       real f

       external f

       real cost

       parameter (cost=64)

       real    hmin, hmax ,k(6), c(21), r

       logical flag

       ifail=0                  

c  Controllo degli estremi

       if (b-a .lt. 0) then

          ifail=23

          goto 999

       endif

c  Controllo del numero iniziale di intervalli

       if (m .le. 0) then

          ifail=24

          goto 999

       endif

c  Controllo della tolleranza assoluta

       if (tol .lt. 0) then

          ifail=2

          goto 999

       endif

c  Controllo del numero massimo di valori da calcolare

       if (n .le. 0) then

          ifail=11

          goto 999

       endif

c  Inizializzazione delle variabili e delle costanti

       i=1

       x(i)=a

       y(i)=y0

       h=(b-a)/m

       hmin=h/cost

       hmax=cost*h

       c(1)=3./8

       c(2)=3./32

       c(3)=9./32

       c(4)=12./13

       c(5)=1932./2197

       c(6)=7200./2197

       c(7)=7296./2197

       c(8)=439./216

       c(9)=3680./513       

       c(10)=845./4104

       c(11)=8./27

       c(12)=3544./2565

       c(13)=1859./4104

       c(14)=11./40

       c(15)=16./135

       c(16)=6656./12825

       c(17)=28561./56430

       c(18)=9./50

       c(19)=2./55

       c(20)=128./4275

       c(21)=2197./75240

       flag=.true.

       if (n .eq. 1  .or.  b-a .eq. 0) goto 999

c  N.B.: la soluzione coincide con la condizione iniziale

c  Ciclo a condizione: termina se il passo h è minore del passo

c  minimo hmin oppure l'ultima ascissa è esterna all'intervallo

   10  continue      

c    Controllo del raggiungimento del numero massimo di valori          

         if (i .eq. n) then

            ifail=25

            goto 999

         endif                                                

c    Calcolo delle costanti K1,..,K6 e della nuova approssimazione

         k(1)=h*f(x(i),y(i))

         k(2)=h*f(x(i)+h/4,y(i)+k(1)/4)

         k(3)=h*f(x(i)+c(1)*h,y(i)+c(2)*k(1)+c(3)*k(2))

         k(4)=h*f(x(i)+c(4)*h,y(i)+c(5)*k(1)-c(6)*k(2)

     %            +c(7)*k(3)) 

         k(5)=h*f(x(i)+h,y(i)+c(8)*k(1)-8*k(2)+c(9)*k(3)

     %            -c(10)*k(4))

         k(6)=h*f(x(i)+h/2,y(i)-c(11)*k(1)+2*k(2)-c(12)*k(3)

     %            +c(13)*k(4)-c(14)*k(5))

         y(i+1)=y(i)+c(15)*k(1)+c(16)*k(3)+c(17)*k(4)

     %         -c(18)*k(5)+c(19)*k(6)

c    Calcolo della differenza tra la soluzione del metodo di ordine 4 

c    e quella del metodo di ordine 5

         r=abs(k(1)/360-c(20)*k(3)-c(21)*k(4)+k(5)/50

     %         +c(19)*k(6))

c    Controllo del raggiungimento della tolleranza richiesta         

         if (r .le. tol*h+tolm(y(i+1))) then

            x(i+1)=x(i)+h

            i=i+1

         endif

c    Calcolo del fattore di proporzionalità

         q=0.84*(tol*h/r)**0.25

c    Modifica del passo h in funzione del valore di q

         if (q .le. 0.1) then

            h=0.1*h

         elseif (q .ge. 4) then

            h=4*h

         else 

            h=q*h

         endif

c    Controllo del nuovo passo ed eventuale modifica                    

         if (h .gt. hmax) h=hmax

c    Controllo del punto di griglia 

         if (x(i) .ge. b) then 

            flag=.false.

         elseif (x(i)+h .ge. b) then

            h=b-x(i)

         elseif (h .lt. hmin) then

            ifail=26

            flag=.false.

         endif   

       if (flag) goto 10

       n=i

  999  continue    

      End

